

Gramática SLR de Tinto

	Reglas
R0	X ::= CompilationUnit
R1	CompilationUnit ::= ImportClauseList TintoDecl
R2	ImportClauseList ::= ImportClauseList ImportClause
R3	ImportClauseList ::= lambda
R4	ImportClause ::= import identifier semicolon
R5	TintoDecl ::= LibraryDecl
R6	TintoDecl ::= NativeDecl
R7	LibraryDecl ::= library identifier lbrace FunctionList rbrace
R8	FunctionList ::= FunctionList FunctionDecl
R9	FunctionList ::= lambda
R10	FunctionDecl ::= Access FunctionType identifier ArgumentDecl FunctionBody
R11	NativeDecl ::= native identifier lbrace NativeFunctionList rbrace
R12	NativeFunctionList ::= NativeFunctionList NetiveFunctionDecl
R13	NativeFunctionList ::= lambda
R14	NativeFunctionDecl ::= Access FunctionType identifier ArgumentDecl semicolon
R15	Access ::= public
R16	Access ::= private
R17	FunctionType ::= Type
R18	FunctionType ::= void
R19	Type ::= int
R20	Type ::= char
R21	Type ::= boolean
R22	ArgumentDecl ::= lparen rparen
R23	ArgumentDecl ::= lparen ArgumentList rparen
R24	ArgumentList ::= Argument
R25	ArgumentList ::= ArgumentList comma Argument
R26	Argument ::= Type identifier
R27	FunctionBody ::= lbrace StatementList rbrace
R28	StatementList ::= StatementList Statement

Gramática SLR de Tinto

	Reglas
R29	StatementList ::= lambda
R30	Statement ::= Decl
R31	Statement ::= IdStm
R32	Statement ::= IfStm
R33	Statement ::= WhileStm
R34	Statement ::= ReturnStm
R35	Statement ::= NoStm
R36	Statement ::= BlockStm
R37	Decl ::= Type IdList semicolon
R38	IdList ::= identifier
R39	IdList ::= identifier assign Expr
R40	IdList ::= IdList comma identifier
R41	IdList ::= IdList comma identifier assign Expr
R42	IfStm ::= if lparen Expr rparen Statement
R43	IfStm ::= if lparen Expr rparen Statement else Statement
R44	WhileStm ::= while lparen Expr rparen Statement
R45	ReturnStm ::= return Expr semicolon
R46	ReturnStm ::= return semicolon
R47	NoStm ::= semicolon
R48	IdStm ::= identifier assign Expr semicolon
R49	IdStm ::= identifier FunctionCall semicolon
R50	IdStm ::= identifier dot identifier FunctionCall semicolon
R51	BlockStm ::= lbrace StatementList rbrace
R52	Expr ::= AndExpr
R53	Expr ::= Expr or AndExpr
R54	AndExpr ::= RelExpr
R55	AndExpr ::= AndExpr and RelExpr
R56	RelExpr ::= SumExpr
R57	RelExpr ::= SumExpr eq SumExpr

Gramática SLR de Tinto

	Reglas
R58	RelExpr ::= SumExpr ne SumExpr
R59	RelExpr ::= SumExpr gt SumExpr
R60	RelExpr ::= SumExpr ge SumExpr
R61	RelExpr ::= SumExpr lt SumExpr
R62	RelExpr ::= SumExpr le SumExpr
R63	SumExpr ::= not ProdExpr
R64	SumExpr ::= minus ProdExpr
R65	SumExpr ::= plus ProdExpr
R66	SumExpr ::= ProdExpr
R67	SumExpr ::= SumExpr minus ProdExpr
R68	SumExpr ::= SumExpr plus ProdExpr
R69	ProdExpr ::= Factor
R70	ProdExpr ::= ProdExpr prod Factor
R71	ProdExpr ::= ProdExpr div Factor
R72	ProdExpr ::= ProdExpr mod Factor
R73	Factor ::= Literal
R74	Factor ::= Reference
R75	Factor ::= lparen Expr rparen
R76	Literal ::= integer_literal
R77	Literal ::= char_literal
R78	Literal ::= true
R79	Literal ::= false
R80	Reference ::= identifier
R81	Reference ::= identifier FunctionCall
R82	Reference ::= identifier dot identifier FunctionCall
R83	FunctionCall ::= lparen rparen
R84	FunctionCall ::= lparen ExprList rparen
R85	ExprList ::= Expr
R86	ExprList ::= ExprList comma Expr